Defining L-functions in GP

A tutorial

B. Allombert

IMB
CNRS/Université de Bordeaux

22/06/2018

This project has received funding from the European Union's Horizon 2020 research and innovation programme under grant agreement N° 676541

Hecke L functions

```
? bnf = bnfinit(a^2+23);
? bnr = bnrinit(bnf, 1);
? bnr.clgp
%3 = [3,[3]]
? Hecke = lfuncreate([bnr,[1]]);
? Hecke[2..5]
? z=lfun(Hecke,0,1)
%4 = 0.28119957432296184651205076406787829979+0.E-6
? algdep(exp(z),3)
%5 = x^3-x-1
```

Galois group

We start with a Galois extension of the rationals, here $\mathbb{Q}(\sqrt[3]{2}, \zeta_3) = \mathbb{Q}(\sqrt[6]{-108})$, with Galois group isomorphic to $\mathfrak{S}_3$.

```
? N = nfinit(x^6+108);
? G = galoisinit(N);
```

G is the Galois group of *N*.

Linear representation

```
? [T,o] = galoischartable(G);
? T~
%4 = [1  1  1]
%    [1  1 -1]
%    [2 -1  0]
```

T is the character table of $G \cong \mathfrak{S}_3$, which is defined over $\mathbb{Z}$. The first character is related to the trivial representation, the second to the signature, and the third to a faithful irreducible representation of dimension 2.
The ordering of the conjugacy classes is given by `galoisconjclasses(G)`.

```
? galoisconjclasses(G)
%4 = [[Vecsmall([1,2,3,4,5,6])],[Vecsmall([3,1,2,6,
```

Artin L-function

We compute the Artin function associated to the 3[rd] character.

```
? L = lfunartin(N,G,T[,3],o);
? lfuncheckfeq(L)
%6 = -127
? L[2..5]
%7 = [0, [0, 1], 1, 108]
? z = lfun(L,0,1)
%8 = 1.3473773483293841009181878914456530463
? algdep(exp(z),3)
%9 = x^3-3*x^2-3*x-1
```

which suggests that this function is equal to a Hecke L-function.

```
? bnr = bnrinit(bnfinit(a^2+a+1),6);
? lfunan([bnr,[1]],100)==lfunan(L,100)
%11 = 1
```

A more interesting example

Let E be the curve $E : y^2 = x^3 - x^2 - 4x + 4$, we build the field $\mathbb{Q}(E[3])$ generated by the coordinates of the points of 3-torsions.

```
? E=ellinit([0,-1,0,-4,4]);
? P=elldivpol(E,3)
%13 = 3*x^4-4*x^3-24*x^2+48*x-32
? Q=polresultant(P,y^2-elldivpol(E,2))
%14 = 27*y^8+2240*y^6-41472*y^4-5308416
? R=nfsplitting(Q)
%15 = y^48-48*y^46+1092*y^44-15028*y^42+138900*y^40
```

This defines a Galois extension of $\mathbb{Q}$ with Galois group $GL_2(\mathbb{F}_3)$.

Non monomial representation

```
? N=nfinit(R); G=galoisinit(N);
? [T,o]=galoischartable(G); T~
%17 = [1,1,1,1,1,1,1,1;
%      1,-1,1,1,-1,1,-1,1;
%      2,0,-1,-1,0,2,0,2;
%      2,0,1,-1,-y^3-y,0,y^3+y,-2;
%      2,0,1,-1,y^3+y,0,-y^3-y,-2;
%      3,-1,0,0,1,-1,1,3;
%      3,1,0,0,-1,-1,-1,3;
%      4,0,-1,1,0,0,0,-4]
? o
%18 = 8
```

$o = 8$ means that the variable y denotes a 8-th root of unity.

Non monomial representation

```
? minpoly(Mod(y^3+y, polcyclo(o,y)))
%19 = x^2+2
```

So the coefficients are in $\mathbb{Q}(\sqrt{-2})$. We use the fourth irreducible representation.

```
? L = lfunartin(N,G,T[,4],o);
? L[2..5]
%21 = [0,[0,1],1,1944]
? lfuncheckfeq(L)
%22 = -127
```

Determinant

```
? dT = galoischardet(G,T[,3],o)
%23 = [1,-1,1,1,-1,1,-1,1]~
? dL = lfunartin(N,G,dT,o); dL[2..5]
%24 = [0,[1],1,3];
```

So L is associate to a modular form of weight 1, level 1944 and Nebentypus $\left(\frac{-3}{\cdot}\right)$.

```
? mf=mfinit([1944,1,-3],0);
? M=mfeigenbasis(mf);
? C=mfcoefs(M[1],100);
? subst(lift(C,y,sqrt(-2))[^1]==lfunan(L,100)
%28 = 1
```

Link to E

We reduce the coefficients of L modulo $1+\sqrt{-2}$ of norm 3.

```
? S = lfunan(L,1000); SE = lfunan(E,1000);
? Smod3 = round(real(S))+round(imag(S)/sqrt(2));
? [(Smod3[i]-SE[i])%3|i<-[1..#Smod3],gcd(i,33)==1]
%31 = [0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,0,...
```

The coefficients of L are congruent to the coefficients of the L-function associated to E modulo $1+\sqrt{-2}$.

lfuntwist

lfuntwist allows to twist an L function by a Dirichlet character. The conductors need to be coprime.

```
? E = ellinit([0,-1,1,-10,-20]);
? G = znstar(5,1);
? L=lfuntwist(E,[G, [1]]);
? lfunan(E,10)
%35 = [1,-2,-1,2,1,2,-2,0,-2,-2]
? lfunan([G, [1]],10)
%36 = [1,I,-I,-1,0,1,I,-I,-1,0]
? lfunan(L,10)
%37 = [1,-2*I,I,-2,0,2,-2*I,0,2,0]
```

lfuntwist

We redefine the curve over $\mathbb{Q}(\zeta_5)$.

```
? nf=nfinit(polcyclo(5,'a));
? E2=ellinit(E[1..5],nf);
? localbitprec(64); lfun(E2,2)
%40 = 1.0543811873412420765
? L2=lfuntwist(E,Mod(4,5));
? lfun(E,2)*lfun(L2,2)*norm(lfun(L,2))
%42 = 1.0543811873410821651289745964738865962
```

Genus-2 curve

For the genus-2 curve $y^2 + (x^3 + 1)y = x^2 + x$:

```
? L=lfungenus2([x^2+x,x^3+1]);
? L[2..5]
%43 = [0,[0,0,1,1],2,249]
? lfun(L,1)
%44 = 0.13154950701147875921340134301217526069
? lfunan(L,10)
%45 = [1,-2,-2,1,0,4,-1,0,4,0]
```

Symmetric power of an elliptic curve

```
? E = ellinit([0,0,1,-7,6]);\\ 5077a1
? L = lfunsympow(E,2); L[2..5]
%47 = [0,[0,0,1],3,25775929]
? lfun(L,2)
%48 = 7.546258095366608923770477395762974673 3
? ellmoddegree(E)
%49 = 1984
```